Algebraic manipulation

Expansion

```
> expand((a+b)^2);
```

$$a^2 + 2\,a\,b + b^2 \tag{1.1}$$

```
> expand((a-b)^2);
```

$$a^2 - 2\,a\,b + b^2 \tag{1.2}$$

```
> expand((a-b)*(a+b));
```

$$a^2 - b^2 \tag{1.3}$$

```
> factor(%);
```

$$(a-b)\,(a+b) \tag{1.4}$$

```
> (exp(lambda*x)*sin(omega*x) + kappa*cos(omega*x))^2;
```

$$\left(e^{\lambda x}\sin(\omega x) + \kappa\cos(\omega x)\right)^2 \tag{1.5}$$

```
> expand(%);
```

$$\left(e^{\lambda x}\right)^2\sin(\omega x)^2 + 2\,e^{\lambda x}\sin(\omega x)\,\kappa\cos(\omega x) + \kappa^2\cos(\omega x)^2 \tag{1.6}$$

Difference of squares

```
> expand((w+x+y+z)^2 - (w+x+y)^2);
```

$$2\,z\,w + 2\,x\,z + 2\,y\,z + z^2 \tag{2.1}$$

```
> 1000000^2 - 999999^2;
```

$$1999999 \tag{2.2}$$

The Cauchy-Schwartz inequality

The Cauchy-Schwartz inequality says that for any real numbers u, v, w, x, y, z we have
$(x\,u + y\,v + z\,w)^2 \le (x^2 + y^2 + z^2)\,(u^2 + v^2 + w^2)$.
In fact we have
$(x^2 + y^2 + z^2)\,(u^2 + v^2 + w^2) = (x\,u + y\,v + z\,w)^2 + (x\,v - y\,u)^2 + (y\,w - z\,v)^2 + (z\,u - w\,x)^2$
$= (x\,u + y\,v + z\,w)^2 +$ some extra, positive terms.

To check this, we give names to the various terms:

```
> A := (x^2+y^2+z^2) * (u^2+v^2+w^2);
```

$$A := (x^2+y^2+z^2)\,(u^2+v^2+w^2) \tag{3.1}$$

```
> B := (x*u+y*v+z*w)^2;
```

$$B := (x\,u+y\,v+z\,w)^2 \tag{3.2}$$

```
> C := (x*v-y*u)^2 + (y*w-z*v)^2 + (z*u-x*w)^2;
```

$$C := (-y\,u+x\,v)^2+(-z\,v+y\,w)^2+(z\,u-w\,x)^2 \tag{3.3}$$

```
> expand(A);
```

$$u^2x^2+u^2y^2+u^2z^2+v^2x^2+v^2y^2+v^2z^2+w^2x^2+w^2y^2+w^2z^2 \tag{3.4}$$

```
> expand(B);
```

$$u^2x^2+2\,u\,v\,x\,y+2\,u\,w\,x\,z+v^2y^2+2\,v\,w\,y\,z+w^2z^2 \tag{3.5}$$

```
> expand(C);
```

$$u^2y^2+u^2z^2-2\,u\,v\,x\,y-2\,u\,w\,x\,z+v^2x^2+v^2z^2-2\,v\,w\,y\,z+w^2x^2+w^2y^2 \tag{3.6}$$

```
> expand(A-B-C);
```

$$0 \tag{3.7}$$

```
> unassign('A','B','C');
```

Factoring

```
> factor(p^2-q^2);
```

$$(p-q)\,(p+q) \tag{4.1}$$

```
> factor(p^3-q^3);
```

$$(p-q)\,(p^2+p\,q+q^2) \tag{4.2}$$

```
> factor(p^4-q^4);
```

$$(p-q)\,(p+q)\,(p^2+q^2) \tag{4.3}$$

```
> factor(a*x^2+b*x^2+a*y^2+b*y^2);
```

$$(x^2+y^2)\,(a+b) \tag{4.4}$$

```
> factor(1+t+t^2+t^3+t^4+t^5+t^6+t^7);
```

$$(t+1)\,(t^2+1)\,(t^4+1) \tag{4.5}$$

```
> x^5-10*x^4+35*x^3-50*x^2+24*x;
```

$$x^5-10\,x^4+35\,x^3-50\,x^2+24\,x \tag{4.6}$$

```
> factor(%);
```

$$x\,(x-1)\,(x-2)\,(x-3)\,(x-4) \tag{4.7}$$

Powers

```
> (-3)^4;
```

$$81 \tag{5.1}$$

```
> ((-3)^4)^(1/4);
```

$$81^{1/4} \tag{5.2}$$

```
> simplify(%);
```

$$3 \tag{5.3}$$

```
> simplify((a^4)^(1/4));
```

$$(a^4)^{1/4} \tag{5.4}$$

```
> simplify((a^4)^(1/4),symbolic);
```

$$a \tag{5.5}$$

The cross-ratio

```
> chi := (a,b,c,d) -> (d-a)*(c-b)/((d-b)*(c-a));
```

$$\chi := (a, b, c, d) \mapsto \frac{(d-a)\,(c-b)}{(d-b)\,(c-a)} \tag{6.1}$$

```
> x := chi(a,b,c,d);
```

$$x := \frac{(d-a)\,(c-b)}{(d-b)\,(c-a)} \tag{6.2}$$

```
> y := chi(1/a,1/b,1/c,1/d);
```

$$y := \frac{\left(\frac{1}{d}-\frac{1}{a}\right)\left(\frac{1}{c}-\frac{1}{b}\right)}{\left(\frac{1}{d}-\frac{1}{b}\right)\left(\frac{1}{c}-\frac{1}{a}\right)} \tag{6.3}$$

```
> simplify(x-y);
```

$$0 \tag{6.4}$$

```
> unassign('chi','x','y');
```

Odd one out

Which of the following is the odd one out? (You should assume everywhere that $x>1$.)

$$a = \frac{(x+1)^2-(x-1)^2}{(x+1)^2-2\left(x^2-1\right)+(x-1)^2}$$

$$b = \frac{x+1+x^{-1}+x^{-2}}{x^{-1}+1+x+x^2}$$

$$c = x^{\frac{1}{x}} \left(x^{x-1}\right)^{\frac{1}{x}}$$

$$d = \sqrt{1 + \sqrt{x-1}\,\sqrt{x+1}\,\sqrt{x^2-1}}$$

```
> a := ((x+1)^2-(x-1)^2)/((x+1)^2-2*(x^2-1)+(x-1)^2);
```

$$a := \frac{(x+1)^2 - (x-1)^2}{(x+1)^2 - 2x^2 + 2 + (x-1)^2} \tag{7.1}$$

```
> b := (x+1+x^(-1)+x^(-2))/(x^(-1)+1+x+x^2);
```

$$b := \frac{x + 1 + \frac{1}{x} + \frac{1}{x^2}}{\frac{1}{x} + 1 + x + x^2} \tag{7.2}$$

```
> c := x^(1/x)*(x^(x-1))^(1/x);
```

$$c := x^{\frac{1}{x}} \left(x^{x-1}\right)^{\frac{1}{x}} \tag{7.3}$$

```
> d := sqrt(1 + sqrt(x-1)*sqrt(x+1)*sqrt(x^2-1));
```

$$d := \sqrt{1 + \sqrt{x-1}\,\sqrt{x+1}\,\sqrt{x^2-1}} \tag{7.4}$$

```
> simplify(a);
```

$$x \tag{7.5}$$

```
> simplify(b);
```

$$\frac{1}{x} \tag{7.6}$$

```
> simplify(c);
```

$$x^{\frac{1}{x}} \left(x^{x-1}\right)^{\frac{1}{x}} \tag{7.7}$$

```
> simplify(c,symbolic);
```

$$x \tag{7.8}$$

```
> simplify(d,symbolic);
```

$$\sqrt{1 + \sqrt{x-1}\,\sqrt{x+1}\,\sqrt{x^2-1}} \tag{7.9}$$

```
> plot(d,x=1..5);
```




```
> combine(d,radical,symbolic);
```

$$\sqrt{1+\sqrt{(x-1)\,(x+1)\,\left(x^2-1\right)}} \tag{7.10}$$

```
> simplify(%,symbolic);
```

$$x \tag{7.11}$$

```
> unassign('a','b','c','d');
```

Collecting terms

```
> X := (u^9*(v-1)-v^9*(u-1)+u-v)/((u-1)*(v-1)*(u-v));
```

$$X := \frac{u^9\,(v-1)-v^9\,(u-1)+u-v}{(u-1)\,(v-1)\,(u-v)} \tag{8.1}$$

```
> Y := simplify(X);
```

$$Y := u^7+(1+v)\,u^6+\left(v^2+v+1\right)u^5+(1+v)\,\left(v^2+1\right)u^4+\left(v^4+v^3+v^2+v+1\right)u^3 \tag{8.2}$$
$$+\left(v^5+v^4+v^3+v^2+v+1\right)u^2+\left(v^6+v^5+v^4+v^3+v^2+v+1\right)u+(1+v)\,\left(v^2\right.$$

$$+1)\,(v^4+1)$$

```
> Y := sort(simplify(X));
```

$$Y := u^7+(v+1)\,u^6+(v^2+v+1)\,u^5+(v+1)\,(v^2+1)\,u^4+(v^4+v^3+v^2+v+1)\,u^3 \quad (8.3)$$
$$+(v^5+v^4+v^3+v^2+v+1)\,u^2+(v^6+v^5+v^4+v^3+v^2+v+1)\,u+(v+1)\,(v^2$$
$$+1)\,(v^4+1)$$

```
> collect(Y,u);
```

$$u^7+(v+1)\,u^6+(v^2+v+1)\,u^5+(v+1)\,(v^2+1)\,u^4+(v^4+v^3+v^2+v+1)\,u^3 \quad (8.4)$$
$$+(v^5+v^4+v^3+v^2+v+1)\,u^2+(v^6+v^5+v^4+v^3+v^2+v+1)\,u+(v+1)\,(v^2$$
$$+1)\,(v^4+1)$$

```
> coeff(Y,u,5);
```

$$v^2+v+1 \quad (8.5)$$

```
> coeff(Y,u^5);
```

$$v^2+v+1 \quad (8.6)$$

```
> coeff(Y,u,0);
```

$$(v+1)\,(v^2+1)\,(v^4+1) \quad (8.7)$$

```
> coeff(Y,u^0);
```

Error, invalid input: coeff received 1, which is not valid for its 2nd argument, x

```
> collect(Y,v);
```

$$v^7+(u+1)\,v^6+(u^2+u+1)\,v^5+(u^3+u^2+u+1)\,v^4+(u^4+u^3+u^2+u+1)\,v^3 \quad (8.8)$$
$$+(u^5+u^4+u^3+u^2+u+1)\,v^2+(u^6+u^5+u^4+u^3+u^2+u+1)\,v+u^7+u^6$$
$$+u^5+u^4+u^3+u^2+u+1$$

```
> coeff(Y,v,0);
```

$$u^7+u^6+u^5+u^4+u^3+u^2+u+1 \quad (8.9)$$

```
> unassign('X','Y');
```